I.T.I.S. "M. PANETTI" - Bari
LABORATORIO DI SISTEMI ELETTRONICI AUTOMATICI

***CLASSE*: 3 ETB**
***ALUNNI*: Lubes Francesco, Tamma Nicola**
Docente: Prof. Ettore Panella

SISTEMI ITERATIVI

Un sistema si dice **iterativo** se può essere descritto da una formula che consente di calcolare il valore futuro conoscendo quello presente. In altre parole lo stato futuro del sistema dipende da quello presente oltre che dai parametri del sistema.

L'analisi dei sistemi con metodi iterativi risulta interessante poiché è possibile impiegare metodi informatici come, ad esempio, il foglio elettronico.

Per comprendere l'importanza di tale tecnica è sufficiente ricordare che i metodi di compressione ed elaborazione dei file audio e video digitali si basano su formule iterative.

Anche lo studio dei **filtri digitali**, noti anche come filtri numerici, si basano su metodi iterativi derivati dall'analisi con la trasformata z.

In questa sede si descrivono alcuni semplici esempi di sistemi iterativi sviluppati con il foglio elettronico.

In generale, indicando con $V(t)$il valore della grandezza in esame al tempo t e con $V(t+\Delta t)$il valore futuro dopo un intervallo Δt si ha:

$$V(t+\Delta t) = V(t) \pm K\cdot V(t)$$

Dove K è un parametro del sistema. Spesso K è una costante che dipende dal problema in esame.

ANALISI DELLA CARICA DI UN CONDENSATORE CON IL METODO ITERATIVO DI EULERO

In fig. 1 si riporta lo schema elettrico di un circuito RC, con uscita sul condensatore, alimentato da una tensione continua V_i = 10 V.



Fig. 1. Circuito RC

Si vuole studiare l'andamento della tensione ai capi del condensatore, supposto inizialmente scarico, quando sottoposto ad una tensione di entrata V_i.

Il metodo impiegato è quello **iterativo di Eulero** che consente di ricavare una formula di facile implementazione sul foglio elettronico.

Ricordiamo che la capacità di un condensatore si può esprimere come rapporto tra la carica accumulata e la tensione ai sui capi:

$$C = \frac{\Delta q}{\Delta V} \qquad [F]$$

Dividendo ambo i membri per Δt si ricava:

$$\frac{\Delta q}{\Delta t} = C \cdot \frac{\Delta V}{\Delta t} \qquad \text{quindi} \qquad i(t) = C \cdot \frac{\Delta V}{\Delta t}$$

Dall'analisi della fig. 1 si ha:

$$\begin{aligned}
v_i(t) &= v_R(t) + v_C(t) \\
v_i(t) &= R \cdot i(t) + v_C(t) \\
v_i(t) &= R \cdot C \cdot \frac{\Delta v_C}{\Delta t} + v_C(t) \\
v_i(t) &= R \cdot C \cdot \frac{v_C(t+\Delta t) - v_C(t)}{\Delta t} + v_C(t) \\
\frac{v_C(t+\Delta t) - v_C(t)}{\Delta t} &= \frac{v_i(t) - v_C(t)}{RC}
\end{aligned}$$

In definitiva si ricava:

$$v_C(t+\Delta t) = v_C(t) + \frac{\Delta t}{RC}\cdot\left[v_i(t) - v_C(t)\right]$$

La precedente formula è nota come **Formula di Eulero**.

È una formula iterativa poiché il valore della tensione v_c al tempo al tempo t + Δt, dipende dal valore al tempo t e dai parametri del sistema. La formula di Eulero si presta ad un facile sviluppo sul foglio elettronico. È sufficiente compilare una tabella con due colonne: una relativa al tempo e l'altra alla tensione ai capi del condensatore ed applicare la formula.

In tabella 1 si mostra il foglio di lavoro di Excel con i dati relativi al circuito in studio.

Tabella 1

	A	B	C	D	E
1					
2	TEMPO	Vc			
3	0,00	0			
4	0,20	2,00		DATI	
5	0,40	3,60		Vi [V]:	10
6	0,60	4,88		R [Ω]:	1,0E+06
7	0,80	5,90		C [F]:	1,0E-06
8	1,00	6,72		τ [s]:	1,0E+00
9	1,20	7,38		Δt [s]:	2,0E-01
10	1,40	7,90			
11	1,60	8,32			
12	1,80	8,66			
13	2,00	8,93			
14	2,20	9,14			
15	2,40	9,31			
16	2,60	9,45			
17	2,80	9,56			
18	3,00	9,65			
19	3,20	9,72			
20	3,40	9,77			
21	3,60	9,82			
22	3,80	9,86			
23	4,00	9,88			
24	4,20	9,91			
25	4,40	9,93			
26	4,60	9,94			
27	4,80	9,95			
28	5,00	9,96			
29	5,20	9,97			
30	5,40	9,98			
31	5,60	9,98			
32	5,80	9,98			
33	6,00	9,99			

In tabella 2 si riporta la struttura del foglio di lavoro nel quale si mostrano le formule impiegate. Ovviamente si devono trascinare la formule fino od ottenere la tensione finale di regime.

Tabella 2

	A	B	C	D	E
1					
2	TEMPO	Vc			
3	0	0			
4	=A3+$E$9	=B3+$E$9/$E$8*($E$5-B3)		DATI	
5	=A4+$E$9	=B4+$E$9/$E$8*($E$5-B4)		Vi [V]:	10
6	=A5+$E$9	=B5+$E$9/$E$8*($E$5-B5)		R [Ω]:	1000000
7	=A6+$E$9	=B6+$E$9/$E$8*($E$5-B6)		C [F]:	0,000001
8	=A7+$E$9	=B7+$E$9/$E$8*($E$5-B7)		τ [s]:	=$E$7*$E$6
9	=A8+$E$9	=B8+$E$9/$E$8*($E$5-B8)		Δt [s]:	=E8/5
10	=A9+$E$9	=B9+$E$9/$E$8*($E$5-B9)			
11	=A10+$E$9	=B10+$E$9/$E$8*($E$5-B10)			

Utilizzando la nota procedura di autocomposizione grafico si ricava.



Fig. 2. Andamento della tensione ai capi del condensatore.

Come descritto dalla teoria si osserva che la tensione ai capi del condensatore parte dal valore iniziale $V_c(0) = 0$ V e tende al valore finale $V_c(\infty) = V_i = 10$ V.

Un parametro che caratterizza la velocità di carica del condensatore è la costante di tempo:

$$\tau = RC$$

Si può verificare, sia teoricamente che sperimentalmente, che la carica del condensatore si può ritenere esaurita dopo un tempo, detto **tempo di assestamento** t_a, pari a:

$$t_a \approx 5 \cdot \tau$$

Infatti, nell'esempio risulta $\tau = 1s$ e il tempo di assestamento è di 5s.

Per concludere si vuole ricordare che, utilizzando altri metodi, la tensione $v_c(t)$ si può esprimere mediante la seguente relazione:

$$v_c(t) = V_i(1 - e^{-t/RC})$$

Tale formula porta, ovviamente, agli stessi risultati ottenuti con il metodo di Eulero.

CALCOLO DELLA VELOCITA' DI CADUTA DI UN GRAVE IN FLUIDO

Si vuole studiare l'andamento della velocità di caduta di un corpo in un fluido impiegando il metodo iterativo. Si deve ricavare l'equazione di stato iterativa che descrive il sistema.
In fig. 3 si mostra una schematizzazione del sistema.



Fig. 3. Rappresentazione schematica della caduta di un grave.

Sul corpo immerso nel fluido agiscono contemporaneamente tre forze: F_P, F_A, F_V.

$F_P = m \cdot g$	**Forza Peso**: è il peso del corpo (m = massa del corpo; g = 9,81m/s^2 accelerazione gravitazionale).
$F_A = d \cdot V \cdot g$	**Forza di Archimede**: si oppone alla F_P ed è uguale al peso del volume del fluido spostato (d = densità de fluido spostato; V = volume del corpo).
$F_V = K \cdot \eta \cdot L \cdot v(t)$	**Forza di Attrito Viscoso**: forza che si oppone allo spostamento del corpo nel fluido (K = coefficiente di forma; η = viscosità del fluido; L = dimensione lineare del corpo).

La forza complessiva F_{tot} è la somma vettoriale delle tre forze:

$$F_{tot} = F_P + F_A + F_V$$

La seconda legge della dinamica afferma che la forza totale deve uguagliare il prodotto della massa del corpo per la sua accelerazione.

$$F_{tot} = m \cdot a$$

La fisica ci dice che l'accelerazione di un corpo è uguale al rapporto della variazione della velocità nel tempo. In formule:

$$a(t) = \frac{\Delta v}{\Delta t}$$

Che si può esprimere nella forma:

$$a(t) = \frac{v(t + \Delta t) - v(t)}{\Delta t}$$

e quindi:

$$F_{tot} = m \cdot \frac{v(t + \Delta t) - v(t)}{\Delta t}$$

Ricordiamo che la forza di attrito viscoso è proporzionale alla velocità del corpo. La costante di proporzionalità dipende dal tipo di fluido e dalle caratteristiche geometriche del corpo.

$$F_V = K_V \cdot v(t)$$

Nel caso di un corpo sferico di raggio R la forza di attrito viscoso segue la legge di Stokes:

$$F_V = (6 \cdot \pi \cdot \eta \cdot R) \cdot v(t)$$

con $K_V = 6 \cdot \pi \cdot \eta \cdot R$

Si ricava:

$$F_{tot} = m \cdot \frac{v(t + \Delta t) - v(t)}{\Delta t} = m \cdot g - d \cdot V \cdot g - K_V \cdot v(t)$$

Dalla precedente relazione si ottiene, finalmente, l'equazione di stato iterativa:

$$v(t + \Delta t) = v(t) + \Delta t \cdot \left[g - \frac{d \cdot V \cdot g}{m} - \frac{K_V \cdot v(t)}{m} \right]$$

Tale formula è di facile implementazione sul foglio elettronico.

A titolo orientativo nella seguente tabella 3 si riportano alcuni valori della viscosità e della densità.

Tabella 3

	VISCOSITÀ η [N · s/m^2]	DENSITÀ d [Kg/m^3]
ARIA	$1{,}84 \cdot 10^{-5}$	1,25
ACQUA	10^{-3}	1000
GLICERINA	1,5	1269

Sul foglio di elettronico si sono compilate due tabelle (vedi tabella 4): una riguardante i dati, l'altra i valori della velocità in funzione al tempo.

La tabella Dati è stata creata per aver a disposizione simultaneamente tutti i dati da inserire nella formula.

Tabella 4

	B	C	D	E	F	G	H
3							
4		TEMPO	VELOCITA'				
5		0,0	0,0		DATI		
6		0,5	0,4		m	10	
7		1,0	0,7		r	0,12	
8		1,5	1,0		d	1269	
9		2,0	1,2		η	1,5	
10		2,5	1,4		g	9,81	
11		3,0	1,6		Δt	0,5	
12		3,5	1,7		Kv	3,391	
13		4,0	1,8		V	0,007	
14		4,5	1,9		Fa	90	
15		5,0	2,0				
16		5,5	2,1				
17		6,0	2,1				
18		6,5	2,2				
19		7,0	2,2				
20		7,5	2,2				
21		8,0	2,2				
22		8,5	2,3				
23		9,0	2,3				
24		9,5	2,3				
25		10,0	2,3				
26		10,5	2,3				
27		11,0	2,3				
28		11,5	2,3				
29		12,0	2,3				
30		12,5	2,3				
31		13,0	2,4				
32		13,5	2,4				
33		14,0	2,4				
34							
35							

Nella tabella TEMPO – VELOCITÀ il valore del tempo si incrementa aggiungendo al valore precedente quello di Δt; la velocità è data dalla formula iterativa di Eulero.
Di seguito si riportata una parte della tabella con le formule risolutive.

Tabella 5

	A	C	D	E	F	G
3						
4		TEMPO	VELOCITA'			
5		0	0		DATI	
6		=C5+$G$11	=D5+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D5)/$G$6))		m	10
7		=C6+$G$11	=D6+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D6)/$G$6))		R	0,12
8		=C7+$G$11	=D7+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D7)/$G$6))		d	1269
9		=C8+$G$11	=D8+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D8)/$G$6))		η	1,5
10		=C9+$G$11	=D9+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D9)/$G$6))		g	9,81
11		=C10+$G$11	=D10+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D10)/$G$6))		Δt	0,5
12		=C11+$G$11	=D11+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D11)/$G$6))		Kv	=6*3,14*($G$9)*$G$7
13		=C12+$G$11	=D12+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D12)/$G$6))		V	=4/3*3,14*($G$7)^3
14		=C13+$G$11	=D13+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D13)/$G$6))		Fa	=$G$8*$G$13*$G$10
15		=C14+$G$11	=D14+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D14)/$G$6))			
16		=C15+$G$11	=D15+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D15)/$G$6))			
17		=C16+$G$11	=D16+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D16)/$G$6))			
18		=C17+$G$11	=D17+$G$11*($G$10-(($G$8*$G$13*$G$10)/$G$6)-(($G$12*D17)/$G$6))			

Dalla tabella precedente utilizzando l'autocomposizione grafico si ricava:



Fig. 4. Velocità di caduta del grave

Dal grafico si osserva che la curva ottenuta tende a stabilizzarsi su un valore costante della velocità.

Il risultato ottenuto è concorde con la "Prima legge della dinamica" o "Principio d'inerzia" che afferma che se su un corpo in moto non agisce nessuna forza il corpo si muove di moto rettilineo uniforme.

CRESCITA DI UNA POPOLAZIONE DI CELLULE

Si vuole studiale la crescita di una popolazione di cellule in una coltura.
Lo studio è condotto ipotizzando tre casi:

1) Risorse illimitate con mortalità nulla;

2) Risorse illimitate con mortalità non nulla;

3) Risorse limitate con mortalità non nulla.

Se indichiamo con T_a il ***tasso di accrescimento***, il numero di cellule della popolazione si può esprimere:

$$N(t+\Delta t)= N(t) + T_a \cdot N(t)$$

Posto T_n il ***tasso di natalità*** e con T_m il ***tasso di mortalità*** si ha, in prima approssimazione:

a) Nel primo caso le cellule crescono e nessuna muore:

$$N(t+\Delta t)= N(t) + T_n \cdot N(t)$$

b) Nel secondo caso si hanno nascite e morti:

$$N(t+\Delta t)= N(t) + (T_n-T_m) \cdot N(t)$$

c) Nel terzo caso il tasso di accrescimento non è costante ma diminuisce all'aumentare della popolazione fino ad annullarsi quando si raggiunge il valore massimo consentito dalle risorse nutritive. Supponendo un decremento lineare, come mostrato in fig.5. si può scrivere:

$$T_a = T_{a\,max} - \frac{Ta\,max}{N\,max} \cdot N(t)$$



Fig. 5

In tabella 6 si riporta il foglio di calcolo con l'applicazione delle formule nei tre casi indicati.

Tabella 6

	A	B	C	D
1	***CRESCITA DI UNA POPOLAZIONE DI CELLULE***			
2				
3	**Tempo**	**Crescita delle cellule nel caso di mortalità zero e risorse illimitate**	**Crescita delle cellule nel caso di mortalità diversa da zero e risorse illimitate**	**Crescita delle cellule nel caso di mortalità diversa da zero e risorse limitate**
4	0	10,00	10,00	10,00
5	1	19,00	16,00	18,10
6	2	36,10	25,60	31,44
7	3	68,59	40,96	50,84
8	4	130,32	65,54	73,34
9	5	247,61	104,86	90,93
10	6	470,46	167,77	98,35
11	7	893,87	268,44	99,81
12	8	1.698,36	429,50	99,98
13	9	3.226,88	687,19	100,00
14	10	6.131,07	1.099,51	100,00
15	11	11.649,03	1.759,22	100,00
16	12	22.133,15	2.814,75	100,00
17	13	42.052,98	4.503,60	100,00
18	14	79.900,67	7.205,76	100,00
19	15	151.811,27	11.529,22	100,00
20				
21		DATI		
22		Numero Iniziale Cellule	10	
23		Tasso Natalità Tn	0,90	
24		Tasso Mortalità Tm	0,30	
25		Numero Massimo di cellule Nmax	100	
26		Tasso di accrescimento massimo Tamax=Tn	0,90	

I seguenti valori si sono ottenuti inserendo nelle relative colonne le formule iterative, come riportato nella tabella 7.

Tabella 7

	A	B	C	D	E
1		*CRESCITA DI UNA POPOLAZIONE DI CELLULE*			
2					
3		**Tempo**	**Crescita delle cellule nel caso di mortalità zero e risorse illimitate**	**Crescita delle cellule nel caso di mortalità diversa da zero e risorse illimitate**	**Crescita delle cellule nel caso di mortalità diversa da zero e risorse limitate**
4		0	10	10	10
5		1	=C4+$D$23*C4	=D4+($D$23-$D$24)*D4	=E4+$D$26*(1-E4/$D$25)*E4
6		2	=C5+$D$23*C5	=D5+($D$23-$D$24)*D5	=E5+$D$26*(1-E5/$D$25)*E5
7		3	=C6+$D$23*C6	=D6+($D$23-$D$24)*D6	=E6+$D$26*(1-E6/$D$25)*E6
8		4	=C7+$D$23*C7	=D7+($D$23-$D$24)*D7	=E7+$D$26*(1-E7/$D$25)*E7
9		5	=C8+$D$23*C8	=D8+($D$23-$D$24)*D8	=E8+$D$26*(1-E8/$D$25)*E8
10		6	=C9+$D$23*C9	=D9+($D$23-$D$24)*D9	=E9+$D$26*(1-E9/$D$25)*E9
11		7	=C10+$D$23*C10	=D10+($D$23-$D$24)*D10	=E10+$D$26*(1-E10/$D$25)*E10
12		8	=C11+$D$23*C11	=D11+($D$23-$D$24)*D11	=E11+$D$26*(1-E11/$D$25)*E11
13		9	=C12+$D$23*C12	=D12+($D$23-$D$24)*D12	=E12+$D$26*(1-E12/$D$25)*E12
14		10	=C13+$D$23*C13	=D13+($D$23-$D$24)*D13	=E13+$D$26*(1-E13/$D$25)*E13
15		11	=C14+$D$23*C14	=D14+($D$23-$D$24)*D14	=E14+$D$26*(1-E14/$D$25)*E14
16		12	=C15+$D$23*C15	=D15+($D$23-$D$24)*D15	=E15+$D$26*(1-E15/$D$25)*E15
17		13	=C16+$D$23*C16	=D16+($D$23-$D$24)*D16	=E16+$D$26*(1-E16/$D$25)*E16
18		14	=C17+$D$23*C17	=D17+($D$23-$D$24)*D17	=E17+$D$26*(1-E17/$D$25)*E17
19		15	=C18+$D$23*C18	=D18+($D$23-$D$24)*D18	=E18+$D$26*(1-E18/$D$25)*E18

Con composizione grafico si ricavano gli andamenti di crescita nei tre casi descritti.



Fig. 5. Crescita di una popolazione di cellule.

ANALISI DI UN DEPOSITO BANCARIO

Si vuole studiare la crescita di un deposito bancario calcolando l'importo maturato in un tempo indeterminato.
Si deposita in banca un capitale iniziale di 1.000 € si desidera sapere l'ammonterà del capitale, dopo un certo tempo supponendo un tasso di interesse del 2% annuo.
Si escludono tutti i possibili prelievi e/o depositi e spese varie

C (Capitale Iniziale)= 1.000 € T_i (tasso d'interesse)= 2%

Il sistema è iterativa ed è descritto dalla seguente formula:

$$C(t+\Delta t)= C(t) + T_i \cdot C(t)$$

La formula iterativa consente lo sviluppo dell'ammontare del capiate nel tempo, come mostrato nella seguente tabella 8.

Tabella 8

	B	C	D	E	F	G	H
4							
5		Tempo	Crescita del capitale nel corso di dieci anni.				
6		0	1000,00				
7		1	1020,00				
8		2	1040,40		DATI		
9		3	1061,21		Capitale iniziale C	1000	
10		4	1082,43		Interesse annuo I	2%	
11		5	1104,08				
12		6	1126,16				
13		7	1148,69				
14		8	1171,66				
15		9	1195,09				
16		10	1218,99				
17							
18							

Si riporta la struttura della tabella con le relative formule di calcolo.

Tabella 9

	B	C	D	E	F	G	H
4							
5		Tempo	Crescita del capitale nel corso di dieci anni.				
6		0	=$G$9				
7		=C6+1	=D6+$G$10*D6				
8		=C7+1	=D7+$G$10*D7		DATI		
9		=C8+1	=D8+$G$10*D8		Capitale iniziale C	1000	
10		=C9+1	=D9+$G$10*D9		Interesse annuo I	0,02	
11		=C10+1	=D10+$G$10*D10				
12		=C11+1	=D11+$G$10*D11				
13		=C12+1	=D12+$G$10*D12				
14		=C13+1	=D13+$G$10*D13				
15		=C14+1	=D14+$G$10*D14				
16		=C15+1	=D15+$G$10*D15				
17							
18							

Con l’autocomposizione grafico si ottiene.



Fig. 6. Crescita del capitale

Il grafico evidenzia l’aumento lineare del capitale nel tempo.